<u>Annotated WinBUGS Program for LVR-HM3 Model 2</u>

Level-1 Model

$$Y_{tij} = \pi_{0ij} + \pi_{1ij}(Grade_{tij} - 7) + \varepsilon_{tij} \qquad \varepsilon_{tij} \sim N(0, \sigma^2) \qquad \text{(Eq.1)}$$

Level-2 Model

$$\pi_{0ij} = \beta_{00j} + \beta_{01j}(ED\_EXPEC_{ij} - \overline{ED\_EXPEC}_{..}) +$$

$$\beta_{02j}(BEHAV\_PBLMS_{ij} - \overline{BEHAV\_PBLMS}_{..}) + r_{0ij} \qquad r_{0ij} \sim N(0, \tau_{\pi 0j}) \qquad \text{(Eq.2a)}$$

$$\pi_{1ij} = \beta_{10j} + \mathrm{Bw}_j(\pi_{0ij} - \beta_{00j}) + \beta_{11j}(ED\_EXPEC_{ij} - \overline{ED\_EXPEC}_{..}) +$$

$$\beta_{12j}(BEHAV\_PBLMS_{ij} - \overline{BEHAV\_PBLMS}_{..}) + r_{1ij} \qquad r_{1ij} \sim N(0, \tau_{\pi 1j}) \qquad \text{(Eq.2b)}$$

$$Cov(r_{0ij}, r_{1ij}) = 0$$

Level-3 Model

$$\beta_{00j} = \gamma_{000} + \gamma_{001}(\overline{HOMERES}_{.j} - \overline{HOMERES}_{..}) + u_{00j} \qquad u_{00j} \sim N(0, \tau_{\beta 00}) \qquad \text{(Eq.3a)}$$

$$\beta_{01j} = \gamma_{010}$$

$$\beta_{02j} = \gamma_{020}$$

$$\beta_{10j} = \gamma_{100} + \mathrm{Bb}^*(\beta_{00j} - \gamma_{000}) + \gamma_{101}(\overline{HOMERES}_{.j} - \overline{HOMERES}_{..}) +$$

$$\gamma_{102}(\overline{TCHEFF}_{.j} - \overline{TCHEFF}_{..}) + u_{10j} \qquad u_{10j} \sim N(0, \tau_{\beta 10}) \qquad \text{(Eq.3b)}$$

$$\beta_{11j} = \gamma_{110}$$

$$\beta_{12j} = \gamma_{120}$$

$$\mathrm{Bw}_j = \mathrm{Bw\_0} + \mathrm{Bw\_1}(\beta_{00j} - \gamma_{000}) + \mathrm{Bw\_2}(\overline{HOMERES}_{.j} - \overline{HOMERES}_{..}) +$$

$$\mathrm{Bw\_3}(\overline{TCHEFF}_{.j} - \overline{TCHEFF}_{..}) + u_{\mathrm{Bw}j} \qquad u_{\mathrm{Bw}j} \sim N(0, \tau_{\mathrm{Bw}}) \qquad \text{(Eq.3c)}$$

$$Cov(u_{00j}, u_{10j}) = 0, \; Cov(u_{00j}, u_{\mathrm{Bw}j}) = 0, \; Cov(u_{10j}, u_{\mathrm{Bw}j}) = \tau_{\beta 10,\mathrm{Bw}}$$

WinBUGS Data Format for Model 2

Model 2 is an extension of Model 1. In particular, Model 2 contains observed predictors at levels 2 and 3. We begin by inspecting the input files for this analysis. As can be seen below, the level-1 file is identical to the level-1 file for the Model 1 analysis. Recall in particular that “stu[]” is a vector of student ID values which link each time series observation with the appropriate student.

As in the case of the first analysis, the level-2 file below contains a vector of school ID values that links each student to the school that he or she is nested in. In addition, it contains two vectors of length I = 2628 that contain, respectively, educational expectation values (“ED_EXPEC[]”) and behavioral problem values (“BEHAV_PBLMS[]”) for each of the students in our sample. These student-level predictors have been centered around their grand means.

Finally, the level-3 file consists of two vectors of length J = 45 that contain the mean home resource values (“mHOMERES[]”) and the mean teacher effort values (“mTEAEFF[]”) for each school in the sample. These school-level predictors have been centered around their grand means.

Note that all of the predictors have been centered around their grand means.

```
## Level-1 File ##
list(T=8585, I=2628, J=45)
Y[]     stu[]  Grade[]
70.040    1      0
69.510    1      1
71.050    1      2
78.480    1      3
59.360    2      0
65.260    2      1
68.890    2      2
72.040    2      3
68.460    3      0
69.430    3      1
 .
 .
 .
40.190   57      0
40.920   57      1
42.720   58      0
45.250   58      1
 .
 .
 .
53.810  113      0
56.290  113      1
 .
 .
 .

53.630  2528     0
56.600  2528     1
61.460  2528     2
70.150  2528     3
 .
 .
 .
57.480  2628     0
61.900  2628     1
```

```
## Level-2 File ##
sch[]  ED_EXPEC[]  BEHAV_PBLMS[]
 1     -0.996      -0.166
 1      1.004      -0.166
 1      1.004      -0.166
 .        .           .
 .        .           .
 .        .           .
 1      1.004      -0.166
 2      1.004      -0.166
 .        .           .
 .        .           .
 2      2.004      -0.166
 .        .           .
 .        .           .
 .        .           .
 .        .           .
 .        .           .
 .        .           .
45      0.004      -0.166
 .        .           .
 .        .           .
45     -0.996      -0.166
END
```

```
## Level-3 File ##
mHOMERES[]  mtTEAEFF[]
  0.049      -1.127
 -0.587      -0.405
   .            .
   .            .
   .            .
   .            .
   .            .
 -0.252       0.706
END
```

67.850	2628	2
61.640	2628	3

END

WinBUGS code for Model 2

```
### T = total number of obs., 8,585; I = total number of students,
2,628; J = total number of schools, 45 ####

model
{
## Level-1 Model ##

for (t in 1: T) {                                          #line 1

 Y[t]~dnorm(mu[t],sigsqinv);                               #line 2

  mu[t] <- pi0[stu[t]] + pi1[stu[t]]*Grade[t];             #line 3
    }

## Level-2 Model ##

for (i in 1:I) {                                           #line 4

 pi0[i] ~ dnorm(estudnts[i],tauinv2s[sch[i]]);             #line 5

 pi1[i] ~ dnorm(estudntg[i],tauinv2g[sch[i]]);             #line 6

  estudnts[i] <- beta00[sch[i]] + gamma010*ED_EXPEC[i] +
                 gamma020*BEHAV_PBLMS[i];                  #line 7

  estudntg[i] <- betagw[sch[i],1] + betagw[sch[i],2]*(pi0[i]-
                 beta00[sch[i]]) + gamma110*ED_EXPEC[i] +
                 gamma120*BEHAV_PBLMS[i];                  #line 8
    }

## Level-3 Model ##

for (j in 1:J) {                                           #line 9

 beta00[j] ~ dnorm(eschs[j],tauinv3s);                    #line 10

 betagw[j,1:2]~dmnorm(eschgw[j,1:2],tauinv3gw[1:2,1:2]);  #line 11

  eschs[j]  <- gamma000 + gamma001*mHOMERES[j];           #line 12

  eschgw[j,1] <- gamma100 + Bb*(beta00[j]-gamma000) +
             gamma101*mHOMERES[j] + gamma102*mTEAEFF[j];  #line 13

  eschgw[j,2] <- Bw0 + Bw1*(beta00[j]-gamma000) +
                 Bw2*mHOMERES[j] + Bw3*mTEAEFF[j];        #line 14
    }
```

```
# prior specification for the fixed effects:

  gamma010 ~ dnorm(0,1.0E-5);                            #line 15
  gamma020 ~ dnorm(0,1.0E-5);                            #line 16

  gamma110 ~ dnorm(0,1.0E-5);                            #line 17
  gamma120 ~ dnorm(0,1.0E-5);                            #line 18

  gamma000 ~ dnorm(0,1.0E-5);                            #line 19
  gamma001 ~ dnorm(0,1.0E-5);                            #line 20

  gamma100 ~ dnorm(0,1.0E-5);                            #line 21
  gamma101 ~ dnorm(0,1.0E-5);                            #line 22
  gamma102 ~ dnorm(0,1.0E-5);                            #line 23

  Bb  ~ dnorm(0,1.0E-5);                                #line 24
  Bw0 ~ dnorm(0,1.0E-5);                                 #line 25
  Bw1 ~ dnorm(0,1.0E-5);                                #line 26
  Bw2 ~ dnorm(0,1.0E-5);                                 #line 27
  Bw3 ~ dnorm(0,1.0E-5);                                 #line 28


# prior specification for the level-1 precision parameter:

  sigsqinv ~ dpar(1,.0001);                              #line 29

  sigmasq <- 1/sigsqinv;                                 #line 30

# prior specification for the level-2 precisions:

for (j in 1:J) {                                         #line 31
  tauinv2s[j] ~ dpar(1.0,.0001);                         #line 32
  tauinv2g[j] ~ dpar(1.0,.0001);                         #line 33

  tauvar2s[j] <- 1/tauinv2s[j];                          #line 34
  tauvar2g[j] <- 1/tauinv2g[j];                          #line 35
        }
# prior specification for the level-3 precisions:

  tauinv3s ~ dpar(1.0,.0001);                            #line 36

  tauvar3s <- 1/tauinv3s;                                #line 37

  tauinv3gw[1:2, 1:2] ~dwish(S[1:2, 1:2 ], 3);           #line 38
  S[1,1] <-  1.395;                                      #line 39
  S[2,2] <-  0.004;                                      #line 40
  S[1,2] <-  0.000;                                      #line 41
  S[2,1] <-  0.000;                                      #line 42

  tauvar3gw[1:2, 1:2] <- inverse(tauinv3gw[,]);          #line 43

}
```

One key difference between Models 1 and 2 is that in Model 2, student initial status at level 2 is now assumed normally distributed with an expected value "estudnts[i]" (line 5), where "estudnts[i]" is set equal to a value that depends upon student i's "ED_EXPEC[i]" and "BEHAV_PBLMS[i]" values (line 7); as can be seen, the fixed effect coefficients for these predictors are, respectively, "gamma010" and "gamma020". As noted in the section on Model 2 results in our paper, β_{00j} in Equation 2a above now represents an adjusted mean initial status parameter for school j. Thus "beta00[sch[i]]" in line 7 now references an adjusted mean initial status value for the school that student i is nested in.

Lines 6 and 8 help us represent the level-2 equation for student growth rates (Equation 2b). In line 6, it can be seen that student rate of change is assumed normally distributed with an expected value of "estudntg[i]", but in contrast to Model 1, we see that "estudntg[i]" is set equal to a value that depends not only on student i's initial status value (i.e., "pi0[i]"), but also on student i's "ED_EXPEC[i]" and "BEHAV_PBLMS[i]" values as well (line 8); the fixed effects connected with these observed covariates are, respectively, "gamma110" and "gamma120" . As discussed in our paper, β_{10j} in Equation 2b represents an adjusted mean growth rate for school j, and so "betagw[sch[i],1]" in line 8 now references an adjusted mean growth rate value for the school that student i is nested in. Furthermore, "betagw[sch[i],2]" references the initial status / rate of change slope for the school that student i is nested in. In addition, Bw_j in Equation 2b now captures the relationship between student initial status and rate of change in school j holding constant student educational expectations and behavioral problems.

Lines 10 and 12 help us represent Equation 3a. As can be seen, the adjusted initial status parameter for school j is assumed normally distributed with an expected value "eschs[j]" (line 10), which is set equal to a value that depends on the mean home resources value for students in school j ("mHomeRes[j]") (line 12).

In line 11, the adjusted mean growth rate and initial status / rate of change slope for school j ("betagw[j,1:2]") are assumed multivariate normally distributed with expected values "eschgw[j,1:2]" and precision matrix "tauinv3gw[1:2,1:2]".

Line 13 sets the expected growth rate for school j ("eschgw[j,1]") equal to a value that depends on the adjusted mean initial status parameter for school j ("beta00[j]"), and, additionally, the mean home resources and mean teacher effort values for school j. Similar to Equation 3b, "gamma100" represents the grand mean growth rate, and "Bb" is the latent variable regression coefficient relating adjusted school mean initial status ("beta00[j]") to adjusted school mean rates of change. Furthermore "gamma101" and "gamma102" are the fixed effect coefficients connected, respectively, with mean home resources and mean teacher effort.

In line 14 it can be seen that we set the expected initial status / rate of change slope for school j ("eschgw[j,2]") equal to a value that depends on the adjusted mean initial status parameter for school j ("beta00[j]"), and on the mean home resource and mean teacher effort values for school j. Similar to Equation 3c, “Bw1” is the latent variable regression coefficient relating adjusted mean initial status (“beta00[j]”) to school initial status / growth rate slopes, and “Bw2” and “Bw3” are the fixed effects connected with the mean home resource and mean teacher effort predictors.

The specification of priors for the fixed effects and precisions in the model is similar to Model 1 (see Endnote 1 in our paper for further details).